USGS-OFR-84-777 USGS-OFR-84-777

UNITED STATES DEPARTMENT OF INTERIOR

GEOLOGICAL SURVEY

HANDTC, A FORTRAN PROGRAM TO CALCULATE INNER-ZONE TERRAIN CORRECTIONS

By

J. B. Spielman and D. A. Ponce

1984

Open-File Report 84-777

Prepared in cooperation with the
Nevada Operations Office
U. S. Department of Energy
(Interagency Agreement DE-AI08-78ET44802)

Menlo Park, California

1984

Copies of this Open-File Report

may be purchased from

Open-File Services Section

Branch of Distribution

U.S. Geological Survey

Box 25425, Federal Center

Denver, Colorado 80225

PREPAYMENT IS REQUIRED

Price information will be published

in the monthly listing

"New Publications of the Geological Survey"

FOR ADDITIONAL ORDERING INFORMATION

CALL: 303 236-7476

 USGS-OFR-84-777

UNITED STATES
DEPARTMENT of INTERIOR
GEOLOGICAL SURVEY

HANDTC, A FORTRAN PROGRAM TO CALCULATE INNER-ZONE TERRAIN CORRECTIONS

By

J. B. Spielman[1] and D. A. Ponce[1]

[1]U.S. Geological Survey, Menlo Park, CA

Contents

Illustrations

Figure

Tables

Table

Abstract

FORTRAN program "Handtc" computes inner-zone terrain corrections for Hayford-Bowie zones C through F and Hammer zones D through H. The terrain corrections are based on the formula for the gravitational attraction of a vertical cylinder. The input file consists of four record types: a station record, a zone label record, a compartment elevation record, and a blank record. The program generates two files: an output file and a print file. The output file lists the station name, total terrain correction, and the various zone terrain corrections in integer format. The print file lists the station name, station elevation, units, method, zone labels, compartment elevations, terrain corrections for each compartment, terrain correction for each zone, and the total terrain correction. The format of the print file is such that it can be quickly scanned and compared to the input file to check for errors in format, zone labels, compartment elevations, or anomalous terrain corrections. The program was developed on a Digital VAX/VMS 11/750 computer and is compatible with ANSI Standard FORTRAN-77.

Introduction

A program 'handtc' to calculate inner-zone terrain corrections for gravity stations has been developed in support of the U.S. Geological Survey's effort to characterize potential radioactive waste storage sites at the Nevada Test Site, Nevada. Determining complete Bouguer anomalies from gravity data necessitates the calculation of terrain corrections to remove the effect of topography surrounding a gravity station.

Generally, there are two manual terrain correction systems in use. Hayford-Bowie (1912) devised a system of dividing the terrain surrounding a gravity station into zones and compartments. In practice, this is accomplished by placing a transparent template of concentric rings and radial lines (fig. 1a) centered at a gravity station on the largest scale topographic map available. The template's circular rings divide the topography into lettered zones and the radial lines divide each zone into equal compartments. As a convenience, compartments may be considered numbered in a clockwise direction beginning from due north to facilitate the recalculation of a compartment elevation. A system of subcompartments was devised (H. W. Oliver, written commun., 1970) based on Bowie's (1917, p. 9-18) subdivided zones to make the Hayford-Bowie system more accurate in areas of steep terrain (fig. 1b). Generally, subcompartments are used when the terrain correction of a zone is greater than 1 milligal.

The second system, devised by Hammer (1939), was modeled after the Hayford-Bowie system of 1912. Hammer designed the system to obtain equidimensional compartments by relating the ratio of the outer and inner radii of a zone to the width of the compartments in that zone (fig. 2). Hammer beleived that the method maximized accuracy while minimizing the total number of compartments.

Modern terrain correction procedures are separated into a three part process: (1) the estimation of the innermost or field terrain correction, (2) the calculation of the inner-zone correction, and (3) the calculation of outer-zone terrain correction by computer analysis.

Since topographic contour maps do not show the detail required near a station, the innermost terrain correction is usually estimated in the field with the aid of templates and charts, or sketched and later calculated in the office. The innermost zones correspond to the Hayford-Bowie A and B zones with an outer radius of 68 m (223 ft), or the Hammer A, B, and C zones with an outer radius of 53 m (175 ft).

Part 2, the inner-zone corrections are calculated by estimating compartment elevations using a transparent template based on either Hayford-Bowie's or Hammer's system of zones (figs. 1 and 2). Compartment elevations are recorded on a terrain correction form, an example of which is shown in figure 3. The inner-zones generally correspond to Hayford-Bowie zones C through F with an outer radius of 2.29 km (7,513 ft), or Hammer zones D through H with an outer radius of 2.61 km (8,578 ft).

The outer-zone terrain correction which extends from the outer radius of the inner-zone correction (part 2) to a radial distance of 166.7 km (Hayford-Bowie zone O) is calculated using a computer program based on a grid of digital terrain data (Plouff, 1977). The starting radius of the computer terrain correction is a function of the accuracy and availability of digital terrain data. The computer terrain correction may start as close as 590 m (outer radius of Hayford-Bowie zone D) from a station provided digital terrain data is available with average elevations estimated at a 1/4-minute geographic grid.

Program "Handtc" calculates inner-zone terrain corrections for Hayford-Bowie zones C through F (including subzones) and Hammer zones D through H. The inner and outer radii and number of compartments for these zones are shown in table 1. The program is based on the formula for the gravitational attraction of a vertical cylinder (Hammer, 1939). The program is designed for any computer compatible with ANSI Standard FORTRAN-77.

Formula

The formula used for calculating inner-zone terrain corrections is exact and based on the gravity effect along the axis at one end of a vertical cylinder (fig. 4a).

$$g_{cylinder} = 2\pi Gd(h + r - (r^2 + h^2)^{\frac{1}{2}}),$$

where

$g_{cylinder}$ = vertical component of gravity,
G = universal gravity constant,
d = density,
r = radius of cylinder,

and

h = height of cylinder.

The formula for the effect of a zone, which is a cylindrical shell, or the difference of two concentric vertical cylinders of the same height (fig. 4b) is

$$g_{zone} = 2\pi Gd(r_2 - r_1 + (r_1^2 + h^2)^{\frac{1}{2}} - (r_2^2 + h^2)^{\frac{1}{2}}),$$

where

r_1 = inner radius of zone,
r_2 = outer radius of zone,

and

h = height of cylinder
= (station elevation) - (compartment elevation).

For a cylindircal sector or compartment (fig. 4b), the gravity effect is

$$g_{compartment} = \frac{2\pi Gd}{n}\left(r_2 - r_1 + (r_1^2 + h^2)^{\frac{1}{2}} - (r_2^2 + h^2)^{\frac{1}{2}}\right),$$

where

n = number of compartments in the zone or cylindrical shell.

The vertical attraction for each compartment is summed to give the terrain effect for each zone. The terrain corrections for each zone are summed and added to the innermost or field terrain correction to obtain the total "hand" terrain correction for each station.

Program Usage

A generalized flow chart of the program is shown in figure 5. The program operates interactively, requesting the input, output, and print file names.

A sample of the input file is shown in table 2 and consists of 4 types of records: (1) a station record, (2) zone label records, (3) compartment elevation records, and (4) a blank record to signify the end of a station.

The station record contains parameters in the following columns:

Column	Format	Item
1-5	a5	Station name
7-12	f6.0	Elevation
14-16	i3	Field terrain correction in 0.01 mGal units
18	a1	Elevation unit: blank or F for feet, M for meters
20-22	a3	Terrain correction system: blank or HAY for Hayford-Bowie, HAM for Hammer.

The second record contains only a zone label. Zone labels must begin in column 1:

1-2	a2	Zone label: C, C1, C2, D, D1, D2, E, E1, E2, F, F1, F2, G, or H.

The third record contains the compartment elevations, each separated by a blank space. Compartment elevations must be in the same units as the station elevation. Compartment elevations are read in 'list-directed' or 'free' format. The number of elevations must equal the number of compartments for a zone (see table 1), unless a slash is used to terminate the input record. A slash may be used when the remaining compartments (or all compartments) have a terrain correction of zero. A slash will set the remaining compartment elevations equal to the station elevation and thus produce a terrain correction of zero. Zone records and associated compartment elevation records are repeated until all zones desired are completed for a station.

The fourth record type, a blank line, must be placed after the last compartment elevation record to terminate the terrain correction for that particular station.

Program Output

The program generates two files, an output file and a print file. The output file (table 3) lists the station name, total terrain correction, and the various zone terrain corrections, to the nearest 0.01 mGal in integer format. The format is:

Column	Format	Item
1-5	a5	Station name
6-11	i6	Total hand terrain correction
12-17	i6	Hayford-Bowie A and B zone field terrain correction
18-23	i6	Hammer A, B, and C zone field terrain correction
24-29	i6	C zone terrain correctin (Hayford-Bowie only)
30-35	i6	D zone terrain correction (either system)
36-41	i6	E zone terrain correction (either system)
42-47	i6	F zone terrain correction (either system)
48-53	i6	G zone terrain correction (Hammer only)
54-59	i6	H zone terrain correction (Hammer only).

The print file (table 4) is a summary for each station. Terrain corrections are expressed in 0.01 mGal units for each zone and compartment, and are printed below the zone label and compartment elevation, respectively. The format is such that the print file can be quickly scanned and compared to the input file to check for errors in format, zone labels, compartment elevations, or anomalous terrain corrections. Totals for the station and zones are also shown in units of mGal.

Program Messages

Error messages occur for the following conditions: (1) error in opening input file, the program prompts the user to 'try again?' (2) error in reading station record, the program lists the station name and terminates, (3) missing station record, program lists the station name and terminates (4) missing or error in compartment elevation record, the program displays station name, zone, number of elements found, and terminates, (5) missing blank record, program lists the station name and terminates.

A warning message results when a compartment elevation is zero. The program lists the station name, zone, and number of compartments that have zero elevation.

References

Bowie, William, 1917, Investigations of gravity and isostasy: U.S. Coast and Geodetic Survey Special Publication No. 40, 196 p.

Hayford, John F., and Bowie, William, 1912, The effect of topography and isostatic compensation upon the intensity of gravity: U.S. Coast and Geodetic Survey Special Publication No. 10, 132 p.

Hammer, Sigmund, 1939, Terrain corrections for gravimeter stations: Geophysics, vol. 4, p. 184-194.

Plouff, Donald, 1977, Preliminary documentation for a FORTRAN program to compute gravity terrain corrections based on topography digitized on a geographic grid: U.S. Geological Survey Open-File Report 77-535, 45 p.

Figure 1A.



Figure 1B.



Figure 1. -- Hayford-Bowie template for (A) 'whole zones' and (B) subzones at a scale of 1:62,500



Figure 2. -- Hammer template at a map scale of 1:62,500.

HAYFORD-BOWIE TERRAIN CORRECTION FORM

PROJECT: MAP: NAME: DATE:

STATION:		ELEV:							AB:				
C (4)	C1 (8)												
	C2 (8)												
D (6)	D1 (12)												
	D2 (12)												
STATION:		ELEV:							AB:				
C (4)	C1 (8)												
	C2 (8)												
D (6)	D1 (12)												
	D2 (12)												
STATION:		ELEV:							AB:				
C (4)	C1 (8)												
	C2 (8)												
D (6)	D1 (12)												
	D2 (12)												
STATION:		ELEV:							AB:				
C (4)	C1 (8)												
	C2 (8)												
D (6)	D1 (12)												
	D2 (12)												

FORM DATE
1984

DEPARTMENT OF INTERIOR
GEOLOGICAL SURVEY

Figure 3. -- Terrain correction form.



Figure 4A.



Figure 4B.

Figure 4. -- Notation for gravity effect of (A) a cylinder, and (B) a cylindrical compartment.



Figure 5. -- Flow chart of program Handtc.

Table 1. -- Zones, radii, and number of compartments for Hayford-Bowie and Hammer systems

Zone	Radii Inner (m)	Radii Outer (m)	Radii Inner (ft)	Radii Outer (ft)	Number of Compartments
			Hayford-Bowie Zones		
C	68	230	223	755	4
D	230	590	755	1,936	6
E	590	1,280	1,936	4,199	8
F	1,280	2,290	4,199	7,513	10
			Hayford-Bowie Subzones with Subcompartments		
C1	68	130	223	427	8
C2	130	230	427	755	8
D1	230	380	755	1,247	12
D2	380	590	1,247	1,936	12
E1	590	870	1,936	2,854	16
E2	870	1,280	2,854	4,199	16
F1	1,280	1,680	4,199	5,512	20
F2	1,680	2,290	5,512	7,513	20
			Hammer Zones		
D	53	170	175	558	6
E	170	390	558	1,280	8
F	390	895	1,280	2,936	8
G	895	1530	2,936	5,018	12
H	1530	2615	5,018	8,578	12

Table 2. -- Example of input file

```
STAT1   43ØØ Ø2Ø F HAM
d
4417 4423.5 443Ø 4436.5 4443 4449
e
453Ø 4545 4559 4573 4586 4599 4612 4624.5
f
4642 4663.5 4684 47Ø3.5 4722 474Ø 4757.5 4747.5
g
4819 4881 4938 4991 5Ø4Ø 5Ø86.5 513Ø.5 5173 5213 5252 529Ø 5327
h
4976 5Ø57 5131 5198 5261 5321 5377 5431 5483 5533 5581 5627.5

STAT2 1757.Ø  Ø2 M HAY
c
1757 1752 /
d
1754 1743 1749 1765 1766 1767

STAT3   2971  Ø4 M
c1
295Ø 295Ø 2955 2955 2955 294Ø 2935 2945
c2
293Ø 2935 293Ø 294Ø 295Ø 291Ø 29ØØ 293Ø
d1
2885 288Ø 2915 288Ø 29Ø5 2935 295Ø 2895 285Ø 283Ø 288Ø 2925
d2
2825 281Ø 2865 283Ø 2915 297Ø 291Ø 2835 277Ø 275Ø 285Ø 2915
e1
28ØØ 279Ø 285Ø 2825 29ØØ 296Ø 29ØØ 284Ø 279Ø 2745 283Ø 29ØØ 292Ø 29ØØ 288Ø 283Ø
e2
276Ø 275Ø 284Ø 2815 2895 297Ø 2875 2835 277Ø 273Ø 282Ø 288Ø 29ØØ 288Ø 282Ø 279Ø
f1
276Ø 275Ø 28ØØ 28ØØ 285Ø 29ØØ 285Ø 28ØØ 275Ø 27ØØ 275Ø 275Ø 276Ø 273Ø 27ØØ 269Ø 266Ø 265Ø 268Ø 27
f2
272Ø 273Ø 276Ø 276Ø 28ØØ 285Ø 282Ø 278Ø 275Ø 274Ø 27ØØ 27ØØ 272Ø 272Ø 27ØØ 269Ø 265Ø 264Ø 267Ø 26

STAT4 6729.5  Ø1 F
c1
6693 6676 6676 6693 67Ø9 6693 6676 67Ø9
c2
6676 6628 6611 6628 666Ø 6644 6628 6693
d1
666Ø 6594 6578 6562 6545 6578 6578 6611 6594 6578 6594 6693
d2
666Ø 6628 6611 6594 6512 6496 6496 6545 6496 6496 6545 6628

STAT5   212Ø  Ø1
c
2135 214Ø 214Ø 2145
d
226Ø 227Ø 2275 2285 23ØØ 2275
e
2415 244Ø 246Ø 248Ø 25ØØ 248Ø 2455 244Ø
f
2625 2675 2715 276Ø 2795 2835 284Ø 283Ø 28ØØ 2775
```

Table 3. -- Example of output file

STAT	TOTAL	AB	ABC	C	D	E	F	G	H
STAT1	668		2Ø		1ØØ	123	121	152	152
STAT2	4	2		Ø	1				
STAT3	4Ø7	4		57	145	95	1Ø7		
STAT4	52	1		19	32				
STAT5	166	1		2	34	54	74		

Table 4. -- Example of print file

```
     LISTING OF COMPARTMENT ELEVATIONS AND TERRAIN CORRECTIONS FROM HANDTC PROGRAM

STATION: STAT1  ELEVATION: 43ØØ.Ø feet    METHOD: Hammer
d          4417  4424  443Ø  4437  4443  4449
   1ØØ      13    15    16    17    19    2Ø
e          453Ø  4545  4559  4573  4586  4599  4612  4625
   123      11    12    13    15    16    17    19    2Ø
f          4642  4664  4684  47Ø4  4722  474Ø  4758  4748
   121      11    12    13    15    16    17    19    18
g          4819  4881  4938  4991  5Ø4Ø  5Ø87  5131  5173  5213  5252  529Ø  5327
   152       5     7     8     9    11    12    13    15    16    17    19    2Ø
h          4976  5Ø57  5131  5198  5261  5321  5377  5431  5483  5533  5581  5628
   152       5     7     8     9    11    12    13    15    16    17    19    2Ø
TOTAL:   6.68   TABC:  Ø.2Ø              TD:  1.ØØ   TE:  1.23   TF:  1.21   TG:  1.52   TH:  1.52

STATION: STAT2  ELEVATION: 1757.Ø meters  METHOD: Hayford-Bowie
c          1757  1752  1757  1757
     Ø       Ø     Ø     Ø     Ø
d          1754  1743  1749  1765  1766  1767
     1       Ø     Ø     Ø     Ø     Ø     Ø
TOTAL:  Ø.Ø4    TAB:  Ø.Ø2   TC:  Ø.ØØ   TD:  Ø.Ø1

STATION: STAT3  ELEVATION: 2971.Ø meters  METHOD: Hayford-Bowie
c1         295Ø  295Ø  2955  2955  2955  294Ø  2935  2945
    21       2     2     1     1     1     4     6     3
c2         293Ø  2935  293Ø  294Ø  295Ø  291Ø  29ØØ  293Ø
    36       4     3     4     2     1     8    1Ø     4
d1         2885  288Ø  2915  288Ø  29Ø5  2935  295Ø  2895  285Ø  283Ø  288Ø  2925
    61       6     6     2     6     3     1     Ø     4    1Ø    13     6     2
d2         2825  281Ø  2865  283Ø  2915  297Ø  291Ø  2835  277Ø  275Ø  285Ø  2915
    84       9    1Ø     5     8     1     Ø     2     8    15    18     6     1
e1         28ØØ  279Ø  285Ø  2825  29ØØ  296Ø  29ØØ  284Ø  279Ø  2745  283Ø  29ØØ  292Ø  29ØØ  288Ø  283Ø
    49       5     6     3     4     1     Ø     1     3     6     9     4     1     Ø     1     2     4
e2         276Ø  275Ø  284Ø  2815  2895  297Ø  2875  2835  277Ø  273Ø  282Ø  288Ø  29ØØ  288Ø  282Ø  279Ø
    46       6     6     2     3     1     Ø     1     2     5     7     3     1     1     1     3     4
f1         276Ø  275Ø  28ØØ  28ØØ  285Ø  29ØØ  285Ø  28ØØ  275Ø  27ØØ  275Ø  275Ø  276Ø  273Ø  27ØØ  269Ø  266Ø  265Ø  268Ø  27ØØ
    53       2     2     2     2     1     Ø     1     2     2     4     2     2     2     3     4     4     5     5     4     4
f2         272Ø  273Ø  276Ø  276Ø  28ØØ  285Ø  282Ø  278Ø  275Ø  274Ø  27ØØ  27ØØ  272Ø  272Ø  27ØØ  269Ø  265Ø  264Ø  267Ø  2695
    53       3     3     2     2     1     1     1     2     2     2     3     3     3     3     3     3     4     5     4     3
TOTAL:  4.Ø7    TAB:  Ø.Ø4   TC:  Ø.57   TD:  1.45   TE:  Ø.95   TF:  1.Ø7

STATION: STAT4  ELEVATION: 6729.5 feet    METHOD: Hayford-Bowie
c1         6693  6676  6676  6693  67Ø9  6693  6676  67Ø9
     6       1     1     1     1     Ø     1     1     Ø
c2         6676  6628  6611  6628  666Ø  6644  6628  6693
    13       1     2     3     2     1     2     2     Ø
d1         666Ø  6594  6578  6562  6545  6578  6578  6611  6594  6578  6594  6693
    17       Ø     1     2     2     2     2     2     1     1     2     1     Ø
d2         666Ø  6628  6611  6594  6512  6496  6496  6545  6496  6496  6545  6628
    16       Ø     Ø     1     1     2     2     2     1     2     2     1     Ø
TOTAL:  Ø.52    TAB:  Ø.Ø1   TC:  Ø.19   TD:  Ø.32

STATION: STAT5  ELEVATION: 212Ø.Ø feet    METHOD: Hayford-Bowie
c          2135  214Ø  214Ø  2145
     2       Ø     1     1     1
d          226Ø  227Ø  2275  2285  23ØØ  2275
    34       4     5     5     6     7     5
e          2415  244Ø  246Ø  248Ø  25ØØ  248Ø  2455  244Ø
    54       5     6     7     8     8     8     7     6
f          2625  2675  2715  276Ø  2795  2835  284Ø  283Ø  28ØØ  2775
    74       5     5     6     7     8     9     9     9     8     8
TOTAL:  1.66    TAB:  Ø.Ø1   TC:  Ø.Ø2   TD:  Ø.34   TE:  Ø.54   TF:  Ø.74

NUMBER OF STATIONS READ:     5
```

Appendix -- Listing of Program Handtc

```
c  PROGRAM TO CALCULATE INNER ZONE TERRAIN CORRECTIONS FOR
c  HAYFORD-BOWIE ZONES C THORUGH F, AND HAMMER ZONES D THROUGH H
c
c  EXPLANATION OF VARIABLES
c  blank:         flags end of old station and start of new one (blank
c                 card)
c  const:         universal gravity constant (elev in meters,tc in hundredths
c                 of mGal)
c  delev:         difference between station and compartment elevations
c  dens:          density 2.67 g/cm3
c  elev:          station elevation
c  f2msq:         conversion factor of feet to meter squared
c  ham:           hay or ham for Hayford-Bowie or Hammer methods
c  iab:           integer form of field tc, Hayford-Bowie ab or Hammer abc
c  itotal:        integer value of total
c  izone:         integer location of zone in array ring
c  mtype:         equals Hayford-Bowie or Hammer for a station in the
c                 print file
c  ncomp:         number of compartments for the current zone
c  nring:         total no. of zones
c  nring2:        number of whole zones
c  nstat:         number of stations processed
c  ostr:          string to contain output record
c  pstr:          string to contain print record
c  stat:          station name
c  total:         total terrain correction for a station
c  ttab:          real value for iab
c  twopi:         pi multiplied by 2
c  twopicd:       2 * pi * gravity constant * the density
c  type:          feet or meters
c  unit1:         f or m
c  zone:          the zone calculated, (i.e. c, c1, c2, d, d1 . . .)
c
c  EXPLANATION OF ARRAY VARIABLES
c  dr(6):         difference of radii
c  elcomp(12):    elev for each compartment
c  frac(2,12):    constant portion of terrain correction formula
c  icomp(2,12):   number of compartments for each zone
c  ielcomp(20):   elcomp in integer form
c  itc(20):       tc in integer form
c  iflag(12):     flags if a zone has been calculated, (-9999. = not
c                 calculated)
c  itzone(6):     integer form of tzone
c  method(2):     Hayford-Bowie or Hammer
c  osp(9):        spacing for output string
c  psp(9):        spacing for print string
c  r1(2,12):      inner radius of zones
c  r1sq(2,12):    squares of r1 (inner radius)
c  r2(2,12):      outer radius of zones
c  r2sq(2,12):    squares of r2 (outer radius)
c  ring(2,12):    the array of zones, lower case
c  ring2(2,12):   the array of zones, upper case
c  tc(12):        tc for each separate compartment
c  tzone(6):      total tc for each zone
c  unit2(2):      feet or meters
c  zlab(9):       zone labels for print string
```

```
c
c
      dimension elcomp(20),icomp(2,12),r1(2,12),r2(2,12),r1sq(2,12),
     &r2sq(2,12),tzone(12),itzone(12),tc(20),itc(20),ielcomp(20),
     &dr(2,12),frac(2,12),iflag(12),osp(9),psp(9)
      character*1 unit1,b1,slash,elstr(132),ans
      character*2 zone,blank,ring(2,12),ring2(2,12)
      character*3 ham
      character*5 stat
      character*6 unit2(2),type,zlab(9)
      character*13 method(2),mtype
      character*32 infile,printfile,outfile
      character*64 ostr
      character*100 pstr
c
c  INITIALIZE  HAYFORD ZONE LABELS, RADII, AND NUMBER OF COMPARTMENTS
      data (ring(1,j),j=1,12)/'c ','d ','e ','f ','c1','c2',
     &'d1','d2','e1','e2','f1','f2'/
      data (ring2(1,j),j=1,12)/'C ','D ','E ','F ','C1','C2',
     &'D1','D2','E1','E2','F1','F2'/
      data (r1(1,j),j=1,12)/68.,230.,590.,1280.,68.,130.,
     &230.,380.,590.,870.,1280.,1680./
      data (r2(1,j),j=1,12)/230.,590.,1280.,2290.,130.,230.,
     &380.,590.,870.,1280.,1680.,2290./
      data (icomp(1,j),j=1,12)/4,6,8,10,8,8,12,12,16,16,20,20/
c
c  INITIALIZE  HAMMER ZONE LABELS, RADII, AND NUMBER OF COMPARTMENTS
c  VARIABLE RANGE FROM 2-6 SO THAT HAYFORD BOWIE
c  D ZONE AND HAMMER D ZONE HAVE THE SAME IZONE
      data (ring(2,j),j=2,6)/'d ','e ','f ','g ','h '/
      data (ring2(2,j),j=2,6)/'D ','E ','F ','G ','H '/
      data (r1(2,j),j=2,6)/175.,558.,1280.,2936.,5018./
      data (r2(2,j),j=2,6)/558.,1280.,2936.,5018.,8578./
      data (icomp(2,j),j=2,6)/6,8,8,12,12/
c
c  INITIALIZE OUTPUT AND PRINT STRINGS (USED TO LEAVE BLANKS
c  IN THE OUPUT AND PRINT FILES WHERE NO TC WAS CALCULATED)
      data osp/24,30,36,42,48,54,6,12,18/
      data psp/27,39,51,63,75,87,1,15,15/
      data zlab/'   TC:','   TD:','   TE:','   TF:','   TG:',
     &          '   TH:','TOTAL:','  TAB:',' TABC:'/
c
c  INITIALIZE VARIABLES
      data b1,blank,slash/' ','  ', '/' /
      data unit2/'feet  ','meters'/
      data method/'Hayford-Bowie','Hammer       '/
c
      nstat=0
      f2msq=.3048*.3048
      const=0.667
      dens=2.67
      twopi=2*3.14159
      twopicd=twopi*const*dens
```

```
c
c  FILE ATTACHMENT
      print 20
   20 format(' TERRAIN CORRECTION PROGRAM!'/)
   50 print 21
   21 format(' TYPE NAME OF INPUT FILE (__.ITC)'/)
      read(5,100) infile
  100 format(a32)
      open(unit=7,file=infile,form='formatted',status='old',
     &blank='zero',err=201)
      print 22
   22 format(1x,'TYPE NAME OF PRINT FILE (__.HPR)'/)
      read(5,101) printfile
  101 format(a32)
      open(unit=8,file=printfile,form='formatted',status='new')
      print 23
   23 format(1x,'TYPE NAME OF OUTPUT FILE (__.HTC)'/)
      read(5,102) outfile
  102 format(a32)
      open(unit=9,file=outfile,form='formatted',status='new')
c
c  CONVERT HAMMER RADII FROM FEET TO METERS
      do 1 i1=2,6
        r1(2,i1)=.3048*r1(2,i1)
        r2(2,i1)=.3048*r2(2,i1)
    1 continue
c
c  CALCULATE RADII SQUARED, DIFFERENCE OF RADII, AND ARRAY FRAC
      kring=12
      nn=1
      do 2 i2=1,2
        do 3 i3=nn,kring
          r1sq(i2,i3)=r1(i2,i3)*r1(i2,i3)
          r2sq(i2,i3)=r2(i2,i3)*r2(i2,i3)
          dr(i2,i3)=r2(i2,i3)-r1(i2,i3)
          frac(i2,i3)=twopicd/icomp(i2,i3)
    3   continue
        kring=6
        nn=2
    2 continue
c
c  WRITE COLUMN HEADINGS FOR PRINT AND OUTPUT FILES
      write(8,103)
  103 format(6x,'LISTING OF COMPARTMENT ELEVATIONS AND
     &TERRAIN CORRECTIONS FROM HANDTC PROGRAM'/)
      write(9,104)
  104 format(2x,'STAT',2x,'TOTAL',3x,'AB',4x,'ABC',3x,'C',
     &5x,'D',5x,'E',5x,'F',5x,'G',5x,'H')
c
c  DO LOOP TO BEGIN A TERRAIN CORRECTION
      do 4 i4=1,1000
```

```
c
c  INITIALIZE TERRAIN CORRECTIONS TO -9999.
c  WHEN FINISHED WITH A STATION, THIS VALUE INDICATES
c  THAT A TC WAS NOT CALCULATED FOR A ZONE
      total=0.0
      do 5 i5=1,12
        tzone(i5)=-9999.
        iflag(i5)=2
    5 continue
c
c  INITIALIZE OUTPUT AND PRINT STRINGS TO ALL BLANK, FOR
c  EACH NEW STATION
      ostr=' '
      pstr=' '
c
c  READ STATION RECORD .
      read(7,105,err=202,end=999) stat,elev,iab,unit1,ham
  105 format(a5,1x,f6.0,1x,i3,1x,a1,1x,a3)
      nstat=nstat + 1
c
c  TEST FOR FEET OR METERS
      if (unit1.eq.'m' .or. unit1.eq.'M') then
        type=unit2(2)
      else
        type=unit2(1)
      end if
c
c  TEST FOR HAYFORD-BOWIE OR HAMMER
      if (ham.eq.'ham' .or. ham.eq.'HAM') then
        mtype=method(2)
        k=2
        nring=6
        nring2=6
      else
        mtype=method(1)
        k=1
        nring=12
        nring2=4
      end if
c
      write(8,106) stat,elev,type,mtype
  106 format(/1x,'STATION: ',a5,2x,'ELEVATION: ',f6.1,1x,a6,
     &2x,'METHOD: ',a13)
c
c  READ NEXT RECORD AND CHECK FOR ZONE LABEL OR BLANK RECORD
   51 read(7,107,end=600) zone
  107 format(a2)
      if (zone.eq.blank) go to 600
c
c  DETERMINE NUMBER OF COMPARTMENTS
      do 6 i6=1,nring
        if (zone.eq.ring(k,i6) .or. zone.eq.ring2(k,i6)) then
          izone=i6
          go to 52
        end if
    6 continue
      go to 203
   52 ncomp=icomp(k,izone)
```

```
c
c  INITIALIZE COMPARTMENT ELEVATIONS TO STATION ELEVATION (IF A
c  SLASH TERMINATES THE INPUT RECORD, REMAINING ELEMENTS WILL EQUAL
c  THE STATION ELEVATION AND HAVE A TC OF ZERO
      do 7 i7=1,20
    7 elcomp(i7)=elev
c
c  READ COMPARTMENT ELEVATION RECORD AS A STRING
      read(7,109) elstr
  109 format(132a1)
c
c  FIND COLUMN OF FIRST ELEVATION AND END OF LAST ELEVATION
      do 8 i8=1,132
    8 if (elstr(i8).ne.b1) goto 53
      goto 205
   53  do 9 i9=132,1,-1
    9 if (elstr(i9).ne.b1) goto 54
      goto 205
c
c  DETERMINE NUMBER OF COMPARTMENT ELEVATIONS
   54 nel=1
      do 10 i10=i8,i9
      if(elstr(i10).eq.b1 .and. elstr(i10-1).ne.b1) nel=nel+1
      if(elstr(i10).eq.slash) goto 55
   10 continue
      if(nel.ne.ncomp) goto 204
   55 backspace 7
      read(7,*,err=205,end=205) (elcomp(j),j=1,ncomp)
c
c  CALCULATE DELTA ELEVATION AND COMPARTMENT TC
      do 11 i11=1,ncomp
        delev=(elev-elcomp(i11))**2
        if(type.eq.unit2(1)) delev=delev*f2msq
        tc(i11)=frac(k,izone)*(dr(k,izone)+sqrt(r1sq(k,izone)+delev) -
     &  sqrt(r2sq(k,izone)+delev))
        tsum=tc(i11)+tsum
   11 continue
      tzone(izone)=tsum
      tsum=0
c
c  WRITE COMPARTMENT ELEVATIONS AND TC'S TO PRINT FILE
      do 12 i12=1,ncomp
        itc(i12)=tc(i12) + 0.5
        if(tc(i12).le.-0.5) itc(i12)=itc(i12)-1
        ielcomp(i12)=elcomp(i12) + 0.5
        if (elcomp(i12).le.-0.5)ielcomp(i12)=ielcomp(i12)-1
   12 continue
      itzone(izone)=tzone(izone) + 0.5
      write(8,108) ring(k,izone),(ielcomp(j),j=1,ncomp)
  108 format(1x,a2,5x,20(1x,i5))
      write(8,110) itzone(izone),(itc(j),j=1,ncomp)
  110 format(4x,i3,20(1x,i5))
```

```
c
c  TEST FOR COMPARTMENTS WITH ZERO ELEVATION
      nzero=0
      do 13 i13=1,ncomp
   13 if (elcomp(i13).eq.0.) nzero=nzero+1
      if(nzero.gt.0) then
        print 111, nzero,stat,zone
        write(8,111) nzero,stat,zone
  111   format(/1x,'**WARNING,',i2,' COMPARTMENT(S) OF ZERO'
     &  ' ELEVATION FOR STATION: ',a5,', ZONE: ',a2)
      endif
c
      go to 51
c
c  SUMMATIONS OF TC'S
  600 ttab=iab
      m=4
      do 14 i14=k,nring2
        m1=m+1
c         EXAMPLE: HAMMER ZONE NOT DONE
          if (mtype.eq.method(2) .and. tzone(i14).eq.-9999.) then
            tzone(i14)=0.0
c         EXAMPLE: C DONE (WHETHER OR NOT SUBZONES ARE DONE), WHOLE ZONE
          else if (tzone(i14).gt.-9999.) then
            tzone(i14)=tzone(i14)
            iflag(i14)=1
c         EXAMPLE: (C NOT DONE) C1 AND C2 DONE, COMPLETE SUBZONE
          else if (tzone(i14+m).gt.-9999. .and. tzone(i14+m1).gt.-9999.)
     &    then
            tzone(i14)=tzone(i14+m) + tzone(i14+m1)
            iflag(i14)=1
c         EXAMPLE: C1 DONE, C2 NOT DONE, ONE SUBZONE DONE
          else if (tzone(i14+m).gt.-9999. .and. tzone(i14+m1).eq.-9999.)
     &    then
            tzone(i14)=tzone(i14+m)
            iflag(i14)=1
c         EXAAMPLE: C1 NOT DONE, C2 DONE, ONE SUBZONE DONE
          else if (tzone(i14+m).eq.-9999. .and. tzone(i14+m1).gt.-9999.)
     &    then
            tzone(i14)=tzone(i14+m1)
            iflag(i14)=1
c
          else
            tzone(i14)=0.0
          endif
        m=m+1
   14 continue
c
c  CONVERT TO INTEGER
      do 15 i15=k,nring2
      itzone(i15)=tzone(i15) + 0.5
   15 continue
```

```
c
c  CALCULATE GRAND TOTAL
      total=0.0
      do 16 i16=k,nring2
        total=tzone(i16) + total
   16 continue
      total=total + ttab
      itotal=total + 0.5
c
c  CALCULATE MGALS
      do 17 i17=k,nring2
        tzone(i17)=tzone(i17)*.01
   17 continue
      total=total*.01
      ttab=ttab*.01
c
c  WRITE SUBTOTALS AND TOTALS TO PRINT AND OUTPUT FILES
c
c  RESET FLAGS FOR TOTAL, AB, AND ABC
      iflag(7)=1
      if (k.eq.1) then
        iflag(8)=1
        iflag(9)=2
      else
        iflag(8)=2
        iflag(9)=1
      endif
c
c  SET TZONE(7), (8), AND (9) TO TOTAL, AB, AND ABC
      tzone(7)=total
      itzone(7)=itotal
      tzone(8)=ttab
      itzone(8)=iab
      tzone(9)=tzone(8)
      itzone(9)=itzone(8)
c
c  WRITE ZONES C-H, TOTAL, AB OR ABC TO OUTPUT AND PRINT
c  STRINGS FOR CALCULATED ZONES ONLY
      do 18 i18=1,9
      if (iflag(i18).eq.1) then
        write(ostr(osp(i18):osp(i18)+5),112) itzone(i18)
        write(pstr(psp(i18):psp(i18)+5),113) zlab(i18)
        write(pstr(psp(i18)+6:psp(i18)+11),114) tzone(i18)
  112   format(i6)
  113   format(a6)
  114   format(f6.2)
      endif
   18 continue
c
c  WRITE STATION NAME TO OUTPUT STRING
      write(ostr(1:5),115) stat
  115 format(a5)
```

```
c
c  WRITE STRINGS TO OUTPUT AND PRINT FILES
      write(9,116) ostr
      write(8,117) pstr
  116 format(1x,a64)
  117 format(1x,a100)
c
c
    4 continue
c
c  DIMENSION OF DO LOOP (1000) EXCEEDED
      goto 999
c
c  ERROR STATEMENTS
  201 print 24
   24 format(/1x,'ERROR IN OPENING INPUT FILE. TRY AGAIN? (Y OR N)')
      read(5,118) ans
  118 format(a1)
      if(ans.eq.'y' .or. ans.eq.'Y') then
        goto 50
      else
        goto 1000
      endif
  202 print 119, stat
      write(8,119) stat
  119 format(/1x,'**STOP, ERROR IN READING STATION RECORD: ', a5)
      goto 999
  203 print 120, stat
      write(8,120) stat
  120 format(/1x,'**STOP, ERROR FOR STATION:  ',a5)
      goto 999
  204 print 121, ncomp,nel,stat,zone
      write(8,121) ncomp,nel,stat,zone
  121 format(/1x,'**STOP, ERROR IN COMPARTMENT ELEVATION RECORD,',
     &' EXPECTING ',i2,' ELEMENTS, FOUND ',i2,','
     &/9x,'FOR STATION: ',a5,', ZONE: ',a2)
      goto 999
  205 print 122, stat,zone
      write(8,122) stat,zone
  122 format(/1x,'**STOP, ERROR IN COMPARTMENT ELEVATION RECORD',
     &' FOR STATION: ',a5,', ZONE: ',a2)
c
c  PRINT STATIONS PROCESSED, CLOSE FILES
  999 print 123, nstat
      write(8,123) nstat
  123 format(/1x,'NUMBER OF STATIONS READ:  ',i4/)
      close(7)
      close(8)
      close(9)
 1000 stop
      end
```